


# Logic Systems and Processors

## *cz:Logické systémy a procesory*


Lecturer: Richard Šusta

[richard@susta.cz](mailto:richard@susta.cz), [susta@fel.cvut.cz](mailto:susta@fel.cvut.cz),  
+420 2 2435 7359

Version: 1.0



CTU-FEE in Prague, CR – subject BE5B35LSP



# Why should we learn VHDL ?

- ❑ To understand modern electronics, we need to know a little about circuits and learn their properties through experiments.
- ❑ However, the clarity of graphical block diagrams decreased with the increasing number of their elements, which already caused the **circuit design crisis** around 1980.
- ❑ To solve it, complex schemas were gradually replaced by text-based description languages, so we will explain one of them, but only its basics.



# VHDL

## VHSIC HDL

Very High-Speed Integrated Circuit HDL

HDL=Hardware Description Language

*for formal description*

*and design of electronic circuits*

- 1981: Designed by the U.S. DoD (Department of Defense) to solve the circuit design crisis. VHDL is based on the Ada language, also developed by the U.S. DoD in 1970.

*Note: ADA is named after Ada Lovelace, the daughter of Lord Byron. She was a mathematician and author of the first computer algorithm.*

- 1983-85: VHDL developed by Intermetrics, IBM and Texas Instruments

- 1986: **Rights transferred to IEEE**, pronounced "Eye-triple-E,"  
(Institute of Electrical and Electronics Engineers)

- The U.S. DoD requires a VHDL description of each supplied ASIC.

- 1987: IEEE standard VHDL 1987

- VHDL-1993, VHDL-2008, VHDL-2019: Revised standards with new features

- but VHDL-1993 is still the most frequent version**

We must use **VHDL 1993** with only some **VHDL 2008** elements:



Quartus Lite supports VHDL 1987, **VHDL 1993**

and some basics elements of **VHDL 2008**

GHDL supports almost full **VHDL 2008**.



## Academic usages




We dare say that the choice of the HDL language is less important than the proper coding style.

With knowledge of one HDL, you quickly master another.



And strict VHDL will better guide beginners to correct hardware descriptions.

● VHDL ● Verilog

## Development companies



## Jobs in the USA



# Description of Combinational Logic via "Dataflow Design Style"

Textbooks:

Circuit Design With VHDL dataflow and structural  
*(96 pages, 20 solved examples and 9 practice tasks)*

Books in DCENET library:

The Designers Guide to VHDL, 910 pages

The Low-Carb VHDL Tutorial, 87 pages

# VHDL Code Domains

## VHDL design styles

**dataflow**

*Concurrent  
statements*

**structural**

*Components and  
interconnects*


**behavioral**

*Sequential statements*

- **Registry**
- **State machines**
- **Testbenches**



*concurrent*



*sequential*

*source code domains*

## VHDL to specification

### VHDL for simulation

*file I/O*

*float 3.1415*

*bit 0, 1*

*integer 1234*

*string "1234"*

*character 'a'*

*boolean*

*implementation in the circuit  
problematic or impossible*

### VHDL for synthesis

types derived from

**std\_logic**



# The last lecture: 9 values of MVL-9 used in VHDL

**Stiff source,**  
i.e., with a **low output impedance**,  
in the case of voltage logic

<b>Forcing 1</b>	<b>'1'</b>
<b>Forcing Unknown</b>	<b>'X'</b>
<b>Forcing 0</b>	<b>'0'</b>

**High impedance (disconnected)**

<b>High Impedance</b>	<b>'Z'</b>
-----------------------	------------

**Design note**

- any value can be here

<b>Don't Care</b>	<b>'-'</b>
-------------------	------------

**Non-stiff (weak) source**  
with **higher output impedance**  
in the case of voltage logic

<i>Weak 1</i>	<i>'H'</i>
<i>Weak Unknown</i>	<i>'W'</i>
<i>Weak 0</i>	<i>'L'</i>

*Wired  
Or/And*

*FPGAs usually cannot implement*

**Simulation reports**

<b>Uninitialized</b>	<b>'U'</b>
----------------------	------------



type std\_ulogic is ('U','X','0','1','Z','W','L','H', '-' );

subtype std\_logic is resolved to std\_ulogic;

connection	'U'	'X'	'0'	'1'	'Z'	'W'	'L'	'H'	'-'
'U'	'U'	'U'	'U'	'U'	'U'	'U'	'U'	'U'	'U'
'X'	'U'	'X'	'X'	'X'	'X'	'X'	'X'	'X'	'X'
'0'	'U'	'X'	'0'	'X'	'0'	'0'	'0'	'0'	'X'
'1'	'U'	'X'	'X'	'1'	'1'	'1'	'1'	'1'	'X'
'Z'	'U'	'X'	'0'	'1'	'Z'	'W'	'L'	'H'	'X'
'W'	'U'	'X'	'0'	'1'	'W'	'W'	'W'	'W'	'X'
'L'	'U'	'X'	'0'	'1'	'L'	'W'	'L'	'W'	'X'
'H'	'U'	'X'	'0'	'1'	'H'	'W'	'W'	'H'	'X'
'-'	'U'	'X'	'X'	'X'	'X'	'X'	'X'	'X'	'X'

- In strongly typed languages, such as VHDL, the matching of the internal type structure is not tested; the type names are only important.
- **Type** definitions create isolated data types not convertible to types derived from a type otherwise named.
- However, there is an option to define a **subtype** that narrows the original type. This will put it in the group of an existing type and it can be converted to its members.

More: [Circuit Design With VHDL dataflow and structural](#)  
Chapter 3.3 on pages 24 to 26.


# 1-bit Types

- *all are enumerated types ("enumerated types")*

- Bit - '0' and '1' - *not suitable for circuit synthesis*
- Boolean - TRUE and FALSE in conditions only
- std\_logic is of type **Multi-Valued Logic**

'1', '0', 'X', 'Z', 'U', '-', 'L', 'H', 'W'

All are type in strict VHDL; they are mutually  
unconvertable by type casts!



```
library ieee;
```

```
use ieee.std_logic_1164.all;
```

- *Access to all names declared in the STD\_LOGIC\_1164 package of the IEEE standard logical data type library and some basic functions.*

```
use ieee.numeric_std.all;
```

- *IEEE recommended library, which contains definitions of types for **signed**, **unsigned**, and **integer** numbers along with arithmetic operations.*

You can insert standard libraries into all VHDL codes.

They don't slow Quartus - they are precompiled.

# Basic Rules for Source Code

- VHDL is not case sensitive, but we should keep capitalization. If we define symbol Data, then referring to it as "data" or "DATA" is wrong style.
- VHDL does not understand diacritics and non ASCII characters, not even in comments.
- Identifier must start with a letter, last character cannot be an underscore, two connected underscores are not allowed.
- All statements end with a semi-colon.
- Comments precede with (--) The rest of line is treated as a comment. Carriage return terminates a comment.
- VHDL 2008 allows /\* \*/ block comments. In VHDL 93, we can comment only lines.

# Reminder

A, B: **in** std\_logic\_vector (7 **downto** 0);

Z: **out** std\_logic\_vector (1 **to** 16);


A, B:

7	6	5	4	3	2	1	0
---	---	---	---	---	---	---	---

**downto** is selected  
for numeric vectors

Z:



1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
---	---	---	---	---	---	---	---	---	----	----	----	----	----	----	----




- **Std\_logic** is one lonely wire.
- **Std\_logic\_vector**  
is a numbered collection of wires. A vector of wires is created as an array (bus) in terms of digital hardware.
- Numbers **signed** and **unsigned**  
are also vectors — but with associated information about their numeric values.  
We will discuss them in the next lecture.









# Reduced toWires18 to toWires4



# VHDL file structure




*At least one architecture  $\equiv$  circuit description*




File: **toWires4.vhd**

entity **toWires4** is

architecture *my* of  
**toWires4** is




# IO modes possible in the port declaration



# IN, OUT, INOUT, BUFFER modes

- **IN**: read-only data input in
- **OUT**: write-only output of outgoing data.  
*In VHDL93 its value cannot be read.*
- **BUFFER**: output as well as OUT,  
but its value is readable in VHDL1993.
- **INOUT**: bidirectional bus wire




BUFFER and INOUT modes require more complex implementations. IN and OUT are preferred ➡


# Concurrent assignment $\leq$ $\equiv$ wire, connection

In what order will the operations be performed?

```
s <= a xor b;  
c <= a and s;
```



```
c <= a and s;  
s <= a xor b;
```



- In the circuit at the same time;
- Although they are executed sequentially in the simulation, a random order is chosen in each iteration to emulate the circuit behavior.



## Inputs and outputs

**Y** : **in** **std\_logic\_vector**(**3** **downto** **0**);

**X3, X2, X1, X0** : **out** **std\_logic**;

## Custom wiring

**X0** <= **Y**(**0**); **X1** <= **Y**(**1**);

**X2** <= **Y**(**2**); **X3** <= **Y**(**3**);



*/\* toWires4 converts bus to 4 wires.*

*It is a simplified form of toWires18 DCE/library \*/*

```
library ieee; use ieee.std_logic_1164.all;
```

```
entity toWires4 is
```

```
port( Y : in std_logic_vector(3 downto 0);  
      X3,X2,X1,X0 : out std_logic);
```

```
end entity;
```

```
architecture rtl of toWires4 is
```


```
begin -- architecture
```

```
    X0<=Y(0); X1<=Y(1); X2<=Y(2); X3<=Y(3);  
end architecture;
```




# Creating a Symbol

- ❖ Be an evocation of the right mouse click on a file in the [Files] tab
- ❖ If the file is actively open, it can also be accessed from the File main menu



Quartus II 64-Bit - C:/SPS/LSP\_VeeqMT2\_pr03/LSP\_VHDL1 - LSP\_VHDL1



# (Register-Transfer Level) RTL Viewer: toWires4

The image shows the Quartus II RTL Viewer interface. The top window displays the Project Navigator with the hierarchy: Cyclone IV E: EP4CE115F29C7 > Demo > toWires4:instW4. A context menu is open over the toWires4:instW4 entity, showing options like Settings..., Set as Top-Level Entity, Locate, Copy, Expand All, Collapse All, Print Hierarchy, Print All Design Files, and Properties. The 'Locate' submenu is expanded, showing options to locate the entity in various editors, with 'Locate in RTL Viewer' selected. The bottom window shows the RTL Viewer for the toWires4:instW4 component, displaying a netlist diagram with inputs KEY[3], KEY[2], KEY[1], KEY[0] and outputs K3, K2, K1, K0. The Netlist Navigator on the left shows the hierarchy: Demo > Instances > toWires4:instW4 > Pins > Input > KEY[0], KEY[1], KEY[2], KEY[3] and Output > K0, K1, K2.

Project Navigator

Entity

Cyclone IV E: EP4CE115F29C7

Demo

toWires4:instW4

Settings... Ctrl+Shift+E

Set as Top-Level Entity Ctrl+Shift+J

Locate

Copy Ctrl+C

Expand All

Collapse All

Print Hierarchy

Print All Design Files

Properties

Locate in Assignment Editor

Locate in Pin Planner

Locate in Chip Planner

Locate in Resource Property Editor

Locate in Technology Map Viewer

Locate in RTL Viewer

Locate in Design File

Quartus II 64-Bit EDA Netlist Writer was successful. 0 errors, 0 warnings

293000 Quartus II Full Compilation was successful. 0 errors, 1209 warnings

RTL Viewer - C:/SPS/LSP\_VeekMT2\_pr03/VeekMT2 - VeekMT2

File Edit View Tools Window Help

Search altera.com

Page Title: Locate Page: 1 of 1

Netlist Navigator

Demo

Instances

toWires4:instW4

Pins

Input

KEY[0]

KEY[1]

KEY[2]

KEY[3]

Output

K0

K1

K2

toWires4:instW4

KEY[3]

KEY[2]

KEY[1]

KEY[0]

K3

K2

K1

K0

LSP

27

```
library ieee; use ieee.std_logic_1164.all;
```

```
entity toWires18 is
```

```
( port ( Y : in std_logic_vector(17 downto 0);  
      X17,X16,X15,X14,X13,X12,X11,X10,X9 :out std_logic;  
      X8,X7,X6,X5,X4,X3,X2,X1,X0 :out std_logic);
```

```
end entity;
```

```
architecture rtl of toWires18 is
```

```
begin -- architecture
```

```
( X0<=Y(0); X1<=Y(1); X2<=Y(2); X3<=Y(3); X4<=Y(4);  
  X5<=Y(5); X6<=Y(6); X7<=Y(7); X8<=Y(8); X9<=Y(9);  
  X10<=Y(10);X11<=Y(11); X12<=Y(12);X13<=Y(13);  
  X14<=Y(14);X15<=Y(15);X16<=Y(16);X17<=Y(17);
```

```
end architecture;
```

```
port ( Y : in std_logic_vector(17 downto 0);  
      X17,X16,X15,X14,X13,X12,X11,X10,X9 :out std_logic;  
      X8,X7,X6,X5,X4,X3,X2,X1,X0 :out std_logic);
```

```
X0<=Y(0); X1<=Y(1); X2<=Y(2); X3<=Y(3); X4<=Y(4);  
X5<=Y(5); X6<=Y(6); X7<=Y(7); X8<=Y(8); X9<=Y(9);  
X10<=Y(10);X11<=Y(11); X12<=Y(12);X13<=Y(13);  
X14<=Y(14);X15<=Y(15);X16<=Y(16);X17<=Y(17);
```

# Schema + RTLViewer: toWires18






# TASKS DIRECTLY SOLVABLE BY MULTIPLEXER





# Multiplexers :-)





# Almost all "Concurrent" commands (connections) dataflow description \*

**<=**

**with-select-when**

**when-else**

*Used  
For*

Concurrent  
assignments

Selective  
assignments

Conditional  
assignments

*implemented by*

wires

**multiplexer**  
selection 1 of  $N=2^m$

cascades  
2-input  
**multiplexers**

\* The list does not include **for-generate** to repeated insertions of codes  
and **process** - sequential code domain.


# With -Select-When

## Selective Signal Assignment

**m-bit** address selects one of the  $N=2^m-1$  inputs

```
with address select  
  target_signal <= expression0 when addr0,  
                  expression1 when addr1,  
                  ...  
                  expressionN_1 when addrN_1;
```

All inputs  
of any  
multiplexer  
must always  
obtain assigned  
values!



- *Conceptually, we can consider the selected signal assignment as a switch that routes the input to the output specified by an address.*
- *All choices must be included, unless the **others** clause is used as the last choice. Ranges and selections can be used as the choice, but it is not allowed for choices to overlap.*
- More: [Circuit Design With VHDL dataflow and structural](#)  
Chapter 3, page 22.

```
bool target_signal, x, y; int address;  
//...some statements  
switch ( address&0x3 )  
{  
    case 0: target_signal = true; break; // expression0  
    case 1: target_signal = (x^y); break; // expression1  
    case 2: target_signal = !y; break; // expression2  
    default: target_signal = false; break; // expressionN_1  
}
```

# Switch vs with..select


```
bool target_signal, x, y; int address;  
//...some statements  
switch (address&0x3)  
{  
    case 0: target_signal = true; break; // expression0  
    case 1: target_signal = (x^y); break; // expression1  
    case 2: target_signal = !y; break; // expression2  
    default: target_signal = false; break; // expressionN_1  
}
```

```
-----  
signal target_signal, x, y : std_logic;  
signal address : std_logic_vector(1 downto 0);  
begin -- architecture-----  
-----
```

**with address select**

```
target_signal <= '1' when "00",  
                 x or y when "01",  
                 not y when "10",  
                 '0' when others; -- default
```

**others** - everything not mentioned so far, in MVL-9 the *address* signal can have 81 values





with address select

```
target_signal <= '1' when "00",
  x or y when "01",
  not x when "10",
  '0' when others; --
```

default

# DCE/library Muxlplexer4x1



## Inputs and outputs

```
X0,X1,X2,X3 : in std_logic;  
sel : in std_logic_vector(1 downto 0);  
Y : out std_logic;
```

## Multiplexer command

with sel select

```
y <= X0 when "00", X1 when "01",  
      X2 when "10",  
      X3 when others;
```

The VHDL keyword **others** specifies all members not yet listed.





```
library ieee; use ieee.std_logic_1164.all;
entity Multiplexer4x1 is
port -- inputs and outputs
( X0, X1, X2, X3 : in std_logic;
  sel : in std_logic_vector(1 downto 0);
  Y : out std_logic);
end entity;
architecture behavioral of Multiplexer4x1 is
begin
with sel select
  y <= X0 when "00", X1 when "01", X2 when "10",
  X3 when others;
end architecture;
```

# DCE/library Muxlplexer4x1bus

Parameter	Value	Type
WBus	7	Signed Integer

Multiplexer4x1bus



The signal names only changed in the entity:

with sel select

y <= Bus0 when "00", Bus1 when "01",  
Bus2 when "10", Bus3 when **others**;

with sel select

y <= X0 when "00", X1 when "01",  
X2 when "10", X3 when **others**;

# Multiplexer4x1 vs Multiplexer4x1bus

entity **Multiplexer4x1bus** is

generic ( *WBus* : natural := 16 ); -- *The width of input buses*

port( **Bus0**, **Bus1**, **Bus2**, **Bus3** : in std\_logic\_vector((*WBUS*-1) downto 0);

sel : in std\_logic\_vector(1 downto 0);

Y : out std\_logic\_vector((*WBUS*-1) downto 0) );

begin --*The consistency check of our generic parameter*

assert *WBus*>0 report "Expected the width of buses *WBus*>0"

severity *ERROR*;

end entity;

=====

entity **Multiplexer4x1** is

port ( **X0**, **X1**, **X2**, **X3** : in std\_logic;

sel : in std\_logic\_vector(1 downto 0);

Y : out std\_logic );

end entity;


```
library ieee;use ieee.std_logic_1164.all;
entity Multiplexer4x1bus is
generic ( WBus : natural := 16 ); -- The width of input buses
port( Bus0, Bus1, Bus2, Bus3 : in std_logic_vector((WBUS-1) downto 0);
      sel : in std_logic_vector(1 downto 0);
      Y : out std_logic_vector((WBUS-1) downto 0));
begin
    assert WBus>0 report "Expected the width of buses WBus>0"
    severity ERROR;
end entity;
architecture behavioral of Multiplexer4x1bus is
begin
    with sel select
        y <= Bus0 when "00", Bus1 when "01", Bus2 when "10",
            Bus3 when others;
end architecture;
```

# to7SegHex written by multiplexer

```
library ieee; use ieee.std_logic_1164.all;
entity to7SegHex IS
    port( X :in std_logic_vector(3 downto 0); -- A(3)-MSB A(0)-LSB hex input
          HEX: out std_logic_vector(6 downto 0) );-- HEX[6..0] output to 7 segment display
end entity;
architecture behavioural of to7SegHex IS
begin
    -- On VEEK-MT2 board, 7segment LEDs are on when its input = '0'
    with X select
        hex <= "1000000" when "0000", "1111001" when "0001", -- 0, 1
               "0100100" when "0010", "0110000" when "0011", -- 2, 3
               "0011001" when "0100", "0010010" when "0101", -- 4, 5
               "0000010" when "0110", "1111000" when "0111", -- 6, 7
               "0000000" when "1000", "0010000" when "1001", -- 8, 9
               "0001000" when "1010", "0000011" when "1011", -- A, b
               "1000110" when "1100", "0100001" when "1101", -- C, d
               "0000110" when "1110", "0001110" when others; -- E, F "1111"
end architecture;
```

# Technology Map Viewer


will show that Quartus has converted the multiplexor into logic equations



# Implementation of Conditional Concurrent Assignment

## When - Else

```
target_signal <= value1 when condition1 else  
                 value2 when condition2 else  
                 ...  
                 valueP_1 when conditionN_1 else  
                 ValueP;
```







- **When-else** statement creates a cascade of Boolean expressions (multiplexers) arranged in descending order, from the highest priority to the lowest - about the output signal definitely the first expression for which the condition is fulfilled.
- The structure of the **when-else** statement implies a priority structure; Boolean expressions are evaluated in the specified order and the earlier one has a higher priority. The first true condition, it sends its value to the output, all the lower ones are already ignored.

# Example: position of the highest switch

SW(0):1, SW(1):2, SW(2):3



```
byte Location; bool sw2, sw1, sw0; -- switch
//...some statements
if (sw2) Location = 3;
else {
    if (sw1) Location = 2;
    else    Location = (sw0 ? 1 : 0);
}
```



```
byte Location; bool sw2, sw1, sw0; -- switch
//...some statements
if (sw2) Location = 3;
else { if (sw1) Location = 2;
      else { Location = (sw0 ? 1 : 0); }
    }
```

---

```
signal Location : std_logic_vector(1 downto 0);
```

```
-----
begin -- architecture-----
    Location <= "11" when SW(2) else
                "10" when SW(1) else
                "01" when SW(0) else "00";
end architecture;
```




From the beginning of electronics till nowadays, many electronic circuits are based on multiplexers, so we should know possible constructions with them.





# Example

Multiplexer 2 to 1



# DCE/library Multiplexer2x1



# Multiplexor 2x1 is a conditional assignment

```
bool x, y, z;  
//...  
target_signal = z ? x && y : x || y;
```

---

Two main ways of describing it:

```
signal target_signal1, target_signal2, x, y, z : std_logic;  
begin -- architecture --  
-- 1. one 2x1 multiplexer from the cascade  
target_signal1 <= x and y when z else x or y;  
-- 2nd multiplexor 2x1  
with z select  
    target_signal2 <= x and y when z,  
                      x or y when others;  
end architecture;
```





```
library ieee;use ieee.std_logic_1164.all;
```

```
entity Multiplexer2x1 is
```

```
port ( X0, X1 : in std_logic;  
       sel : in std_logic;  
       Y : out std_logic);
```

```
end entity;
```


```
architecture behavioral of Multiplexer2x1 is
```

```
begin
```

```
    Y <= X1 when sel else X0;
```

```
    -- Y <= X1 when sel='1' else X0; -- VHDL93
```

```
end architecture;
```




Both options lead to the same wiring

```
signal target_signal1, target_signal2, x, y, z : std_logic;  
begin -- architecture --
```

```
    target_signal1 <= x and y when z else x or y;
```

```
    with z select
```

```
        target_signal2 <= x and y when z,  
                        x or y when others;
```



# The **signal** is not a **variable**, but a **conductor**!

```
signal target_signal, x, y, z : std_logic;
```


```
begin -- architecture --
```

```
→ target_signal <= x and y when z else x or y;
```

```
with z select -- redefine first value --
```

```
→ target_signal <= x and y when z,  
x or y when others;
```



```
end architecture;
```



---


# DEMUX FROM MUX

# DCE/library Demux1to2 by Mux



```
library ieee; use ieee.std_logic_1164.all;
entity Demux1to2 is
port ( Data : in std_logic; -- data input
      sel : in std_logic; -- selection of output
      X0, X1 : out std_logic);
end entity;
architecture behavioral of Demux1to2 is
begin
    X0 <= '0' when sel else data;
    X1 <= data when sel else '0';
end architecture;
```

*The Demux code is one description from many other possibilities.*



---

# **MUX OF BUSES**

```

library ieee; use ieee.std_logic_1164.all;
entity Multiplexer2x1bus is
generic ( WBUS : natural := 16 ); -- WIDTH (LENGTH) of data Bus
port (Bus0, Bus1 : in std_logic_vector((WBUS-1) downto 0);
      sel : in std_logic;
      Y : out std_logic_vector((WBUS-1) downto 0));
begin
    assert WBus>0 report "Excepted the width of buses WBus>0"
    severity ERROR;
end entity;
architecture behavioral of Multiplexer2x1bus is
begin
    y <= Bus1 when sel else Bus0;
end architecture;


```



# Multiplexor 2x1 vs Buses

```
library ieee; use ieee.std_logic_1164.all;  
entity Multiplexer2x1 is port ( X0, X1 : in std_logic; sel : in std_logic;  
                                Y : out std_logic);
```

```
end entity;  
architecture behavioral of Multiplexer2x1 is  
begin Y <= X1 when sel else X0;  
end architecture;
```




```
library ieee; use ieee.std_logic_1164.all;  
entity Multiplexer2x1bus is  
generic ( WBUS : natural := 16 ); -- WIDTH (LENGTH) of data Bus  
port ( Bus0, Bus1 : in std_logic_vector((WBUS-1) downto 0); sel : in std_logic;  
      Y : out std_logic_vector((WBUS-1) downto 0));  
begin assert WBus > 0 report "Excepted the width of buses WBus > 0" severity ERROR;  
end entity;  
architecture behavioral of Multiplexer2x1bus is  
begin y <= Bus1 when sel else Bus0;  
end architecture;
```

# DCE/library Multiplexer2x1bus

Parameter	Value	Type
WBUS	7	Signed Integer

Multiplexer2x1bus



*we add unconnected SW  
to continuous used series*

Bus0


Bus1

sel

Y



Type	ID	Message
Warning	21074	Design contains 3 input pin(s) that do not drive logic
	15610	No output dependent on input pin "SW[16]"
	15610	No output dependent on input pin "SW[15]"
	15610	No output dependent on input pin "SW[14]"




**y <= Bus1 when sel else Bus0;**



Example


Up Down on 7-segment





Parameter	Value	Type
N	1000001	String
WBus	7	Signed Integer

toConstFromBinary




iUp

Up

Parameter	Value	Type
N	0100001	String
WBus	7	Signed Integer

toConstFromBinary




iDown

down

Parameter	Value	Type
WBUS	7	Signed Integer

Multiplexer2x1bus




sel

iMux

OUTPUT

HEX0[6..0]

SW[0]



INPUT  
VCC

# Port map

---

- Our simple block diagram could be written directly in VHDL, as we show at the end, but we will use it to demonstrate creating instances in VHDL.

**component** Multiplexer2x1bus **is**

generic ( *WBUS* : *natural* := 16 ); -- *WIDTH (LENGTH) of data Bus*

**port** (Bus0, Bus1 : in *std\_logic\_vector*((*WBUS*-1) *downto* 0));

sel : in *std\_logic*;

Y : out *std\_logic\_vector*((*WBUS*-1) *downto* 0) );

**end component**;



library ieee; use ieee.std\_logic\_1164.all;

**entity** Multiplexer2x1bus **is**

generic ( *WBUS* : *natural* := 16 ); -- *WIDTH (LENGTH) of data Bus*

**port** (Bus0, Bus1 : in *std\_logic\_vector*((*WBUS*-1) *downto* 0);

sel : in *std\_logic*;

Y : out *std\_logic\_vector*((*WBUS*-1) *downto* 0));

**begin**

assert *WBus*>0 report "Excepted the width of buses *WBus*>0"

severity *ERROR*;

**end entity**;



```

library ieee; use ieee.std_logic_1164.all;
entity UpDown_VHDL is port ( SW : in std_logic_vector(0 downto 0):="0";
                           HEX0 : out std_logic_vector(6 downto 0));
end entity;
architecture behavioral of UpDown_VHDL is
  component Multiplexer2x1bus is
    generic ( WBUS : natural := 16 ); -- WIDTH (LENGTH) of data Bus
    port -- inputs and outputs
      ( Bus0, Bus1 : in std_logic_vector((WBUS-1) downto 0);
        sel      : in std_logic;
        Y        : out std_logic_vector((WBUS-1) downto 0));
  end component;
begin
  iMux : Multiplexer2x1bus generic map(7)
    port map("1000001", "0100001", SW(0), HEX0);
end architecture;

```





*More reliable **named associations**  
can be given in any order*

```
iMux : Multiplexer2x1bus
  generic map(WBUS => 7)
  port map(sel => SW(0), Y => HEX0,
           Bus0 => "1000001",
           Bus1 => "0100001");
```

Shorter **positional associations**  
must exactly follow the entity order

```
iMux : Multiplexer2x1bus
  generic map( 7 )
  port map("1000001", "0100001", SW(0), HEX0);
```

```

architecture behavioral2 of UpDown_VHDL is
component Multiplexer2x1bus is
    generic ( WBUS : natural := 16 ); -- WIDTH (LENGTH) of data Bus
    port -- inputs and outputs
    (   Bus0, Bus1  : in std_logic_vector((WBUS-1) downto 0);
        sel   : in std_logic;
        Y     : out std_logic_vector((WBUS-1) downto 0));
end component;
begin iMux : Multiplexer2x1bus generic map(WBUS => 7)
    port map(sel => SW(0), Y => HEX0,
            Bus0 => "1000001", Bus1 => "0100001");
end architecture;

```

```
library ieee; use ieee.std_logic_1164.all;
entity UpDown_VHDL is
    port ( SW : in std_logic_vector(0 downto 0):="0";
          HEX0 : out std_logic_vector(6 downto 0));
end entity;

architecture behavioral3 of UpDown_VHDL is
begin
    HEX0 <= "1000001" when SW(0)='0' else
            "0100001";
end architecture;
```

As a top-level entity, it connects to Veek-MT2 pins.

# Technology Map Viewer

Technology Map Viewer - Post-Fitting - C:/SPS2024L/LSPIntroTasks\_Q20/IntroTasks - IntroTasks


File Edit View Tools Window Help

Search altera.com

Page: 1 of 1

Display

IntroTasks:1 +



# Differences **std\_logic** vs **std\_logic\_vector**

As top-level entity, it does not directly use SW Veek-MT2 pin.

```
library ieee;  
use ieee.std_logic_1164.all;  
entity UpDownHex0 is  
    port (SW : in std_logic := '0';  
          HEX0 : out std_logic_vector(6 downto 0));  
end entity;  
architecture behavioral3 of UpDownHex0 is  
begin  
    HEX0 <= "1000001" when SW='0' else  
            "0100001";  
end architecture;
```

